
TP 2 - Méthodes de Monte-Carlo - Corrigé succinct

Exercice 1. 1. L’intégrale I1 est l’aire du disque unité de R2, I2 aussi, et I3 le volume de la boule
unité dans R3. Les intégrales I1 et I2 valent donc π, et I3 vaut 4/3π (et donc π = 3I3/4). On peut
écrire

I1 = 4E[
√

(1− U2)] avec U ∼ U[0;1],
I2 = 4E[1U2

1+U
2
2≤1] avec U1, U2 ∼ U[−1;1], U1 indépendante de U2.

I3 = 8E[1U2
1+U

2
2+U

2
3≤1] avec U1, U2, U3 ∼ U[−1;1], U1, U2, U3 indépendantes.

2. Voici les trois fonctions demandées, enregistrées dans un fichier TP2Ex1CalculPi.sci. On approche
π en utilisant les formules

IN,1 =
4

N

N∑
i=1

√
1− U2

i , IN,2 =
4

N

N∑
i=1

1U2
1,i+U

2
2,i≤1, IN,3 =

8

N

N∑
i=1

1U2
1,i+U

2
2,i+U

2
3,i≤1.

function pi1=TP2Ex1CalculPi_MC1(n)
U=rand(1,n);
pi1=mean(4*sqrt(1-U.^2));

endfunction

function pi2=TP2Ex1CalculPi_MC2(n)
U1=2*rand(1,n)-1;
U2=2*rand(1,n)-1;
pi2=mean(4*bool2s((U1.^2+U2.^2)<=1));

endfunction

function pi3=TP2Ex1CalculPi_MC3(n)
U1=2*rand(1,n)-1;
U2=2*rand(1,n)-1;
U3=2*rand(1,n)-1;
pi3=2*3*mean(bool2s((U1.^2+U2.^2+U3.^2)<=1));

endfunction

3. Soit toujours IN,1 =
∑N
i=1 4

√
1− U2

i /N , pour {U1, . . . , Un} un échantillon de variables U[0;1]. On
cherche N tel que

P
(
|IN,1 − π| ≤ 10−2/2

)
≥ 1− α,

avec α = 0.05.
Le Théorème Central limite et la méthode appliquée dans l’Exercice 5 entraînent

lim
N→∞

P

|IN,1 − π| ≤ q1−α/2
√

Var(4
√
1− U2

1 )

N

 ≥ 1− α.

En minorant grossièrement π (inconnu !) par 3,

Var
(
4
√
1− U2

1

)
=

32

3
− π2 ≤ 5

3
,

et donc on cherche N tel que

2

√
5/3

N
q1−α/2 ≤ 10−2.

On choisit donc N ≥ 4×104× q21−α/2×5/3. On peut “vérifier” dans Scilab que pour un N plus grand
que ce seuil l’écart entre IN,1 et π est bien inférieur à 10−2 (voir le début du script à la question
suivante).
4. Voici un script illustrant la convergence des trois méthodes en représentant les approximations IN,l
en fonction de N .

//Script TP2 Ex 1
clear
exec(’TP2Ex1CalculPi.sci’)

n=1000
pi1=TP2Ex1CalculPi_MC1(n)
pi2=TP2Ex1CalculPi_MC2(n)
pi3=TP2Ex1CalculPi_MC3(n)

N=floor(4*10^4*cdfnor(’X’,0,1,0.975,0.025)^2*5/3)+1; //valeur calculee Question 2.
pi1bis=TP1Ex6CalculPi_MC1(n)
abs(pi1bis-%pi)

//Illustration de la convergence des 3 methodes

vect_n=[100:500:10000];
vect_pi1=[];
vect_pi2=[];
vect_pi3=[];
for n=vect_n

vect_pi1=[vect_pi1,TP2Ex1CalculPi_MC1(n)];
vect_pi2=[vect_pi2,TP2Ex1CalculPi_MC2(n)];
vect_pi3=[vect_pi3,TP2Ex1CalculPi_MC3(n)];

end
scf(1)
clf
plot2d(vect_n,[vect_pi1’,vect_pi2’,vect_pi3’],style=[1,2,3],...

rect=[0,3,10000,3.3])
plot2d([min(vect_n),max(vect_n)],[%pi,%pi],style=5,rect=[0,3,10000,3.3])
legend([’MC1’;’MC2’;’MC3’;’vraie_valeur’]);
xtitle(’Approximation de ’+string(%pi)+’ par Monte Carlo’,’n’,’’)

Le graphe obtenu est représenté à la Figure 1 ci-dessous. On peut aussi illustrer plus en détail la
convergence en représentant IN,l (l = 1, 2, 3) en fonction de N , mais aussi les bornes de l’intervalle
de confiance obtenu par Théorème Central limite pour chaque valeur de N . A titre d’exemple, pour
IN,1, on ajoute donc la fonction suivante au fichier TP2Ex1CalculPi.sci.

function [piInf, piMoy, piSup]=TP2Ex1CalculPi_MC1bis(n)
//meme principe que la 1ere fonction mais retourne les approximations
//obtenues pour toute valeur <=n, et les bornes de l’ICA associe

U=rand(1,n);
piMoy=cumsum(4*sqrt(1-U.^2))./[1:n];
deviation=1.96*sqrt(5/3)./sqrt([1:n]);
piInf=piMoy-deviation;
piSup=piMoy+deviation;

endfunction

et on peut l’utiliser en complétant le script précédent par

//Illustration de la convergence pour la premiere methode,
//avec Bornes de l’IC correspondant (obtenu par TCL)

n=10000;
[piInf, piMoy, piSup]=CalculPi_MC1bis(n)
scf(2)
clf

plot2d([1:n]’,[piInf’,piMoy’,piSup’],style=[1,3,1],rect=[0,3,10000,3.3])
plot2d([min(vect_n),max(vect_n)],[%pi,%pi],style=5,rect=[0,3,10000,3.3])
xtitle(’Approximation de ’+string(%pi)+’ par Monte Carlo1, et bornes d’’ICA’,’n’,’’)

ce qui permet d’obtenir la seconde partie de la Figure 1.

(a) (b)

Figure 1 – Exercice 1. Approximation de π par Monte-Carlo. (a) Convergence des 3 approximations
IN,l, l = 1, 2, 3. (b) Convergence de IN,1 et borne de l’intervalle de confiance de niveau 95%.

Exercice 2. 1. Les variables X et 1 − X ont même loi. Donc I = E[f(X)] = E[f(1 − X)], puis
I = E

[
1
2 (f(X) + f(1−X))

]
.

2. En utilisant l’expression de la variance d’une somme, le fait que X et 1 − X ont même loi, et
l’Inégalité de Cauchy-Schwarz,

Var(g(X)) =
1

4
(Var(f(X)) +Var(f(1−X)) + 2Cov(f(X), f(1−X))) ,

=
1

2
(Var(f(X)) + Cov(f(X), f(1−X))) ≤ Var(f(X)).

3.

// Script TP2 Ex2
n=1000;
vraie_valeur=exp(1)-1;

u1=rand(1,n);
approx1=mean(exp(u1));
erreur1=abs(approx1-vraie_valeur)
disp(’Erreur avec Monte-Carlo classique:’+string(erreur1))

u2=rand(1,n);
approx2=mean((1/2)*(exp(u2)+exp(1-u2)));
erreur2=abs(approx2-vraie_valeur)
disp(’Erreur avec reduction de variance:’+string(erreur2))

Exercice 3. Rappel : si X ∼ P(λ), E[X] = Var(X) = λ.
1. Comme les variables (Xi)i sont i.i.d., L2, d’espérance et de variance λ, le Théorème Central limite
prouve que √

n
(
Xn − λ

) L−→ Zλ ∼ N (0, λ). (1)

Par ailleurs, la Loi faible des grands nombres justifie que (Xn)n converge en probabilité vers
E[X1] = λ. Donc, comme la fonction x 7→

√
x est continue,√
Xn

P−→
√
λ. (2)

Donc par le Lemme de Slutzky, en tenant compte de (1) et (2),
√
n
(
Xn − λ

)√
Xn

L−→ Zλ√
λ
= N ∼ N (0, 1).

Pour bâtir În,1(α), on peut donc chercher q tel que

lim
n→∞

P

(
√
n

∣∣∣∣∣Xn −
λ√
Xn

∣∣∣∣∣ ≥ q
)

= P (|N | ≥ q) ≤ α.

Il suffit de prendre q = q1−α/2 le quantile d’ordre 1− α/2 de la loi N (0, 1). Donc,

În,1(α) =

[
Xn −

q1−α/2√
n

√
Xn;Xn +

q1−α/2√
n

√
Xn

]
.

Remarque : On définit la variance empirique par S2
n =

∑n
i=1(Xi−Xn)

2/(n−1). On peut alors montrer
que l’on a aussi la convergence de (S2

n)n vers Var(X1) = λ en probabilité. On peut alors appliquer la
même méthode : en divisant (1) par

√
S2
n, on obtient une nouvelle convergence en loi vers N , et on

en déduit un second intervalle de confiance asymptotique :

Îbisn,1(α) =

[
Xn −

q1−α/2√
n

√
S2
n;Xn +

q1−α/2√
n

√
S2
n

]
.

2. Par Loi faible des grands nombres et continuité de la fonction x 7→
√
x+
√
λ,√

Xn +
√
λ

P−→ 2
√
λ.

En utilisant ceci et (1), on peut à nouveau appliquer le Lemme de Slutzky pour obtenir

2
√
n

(√
Xn −

√
λ

)
= 2
√
n

(
Xn − λ

)√
Xn +

√
λ

L−→ 2Zλ
2λ

= N ∼ N (0, 1).

Comme ci-dessus, on en déduit un intervalle de confiance asymptotique pour
√
λ, puis, pour n assez

grand, un intervalle de confiance asymptotique pour λ :

În,2(α) =

[(√
Xn −

q1−α/2

2
√
n

)2

;

(√
Xn +

q1−α/2

2
√
n

)2
]
.

3. Soit un rang n fixé. On cherche pour chaque intervalle În à approcher par Monte-Carlo l’espérance
de Y , avec Y = 1λ∈În . On doit donc simuler un N−échantillon (Y1, . . . , YN ) et en prendre la moyenne
empirique. Simuler une variable Y , c’est déjà simuler un n−échantillon de variables Xi ∼ P(λ). On
doit donc, pour chaque valeur de n, simuler un tableau n×N de variables de loi P(λ).

// Script TP2 Ex 3
clear
vect_n=[10,30,50,100,500] //rangs auxquels on teste les niveaux des IC
N=1000 //nb de repetition pour le Monte-Carlo
lambda=3; //parametre de la loi de Poisson

quantile=cdfnor(’X’,0,1,0.975,0.025) //quantile d’ordre 1-alpha/2 de N(0,1)
//avec alpha=0.05

//vecteurs qui vont contenir les niveaux des IC pr les valeurs de n dans vect_n
niveauIC1=[]
niveauIC2=[]
niveauIC1bis=[]

for n=vect_n
Xech=grand(n,N,’poi’,lambda)
Xbar=mean(Xech,’r’) //echantillon de taille N de X_nbar (moy empirique)
S2=variance(Xech,’r’) //echantillon de taille N de S2 (var empirique)

//niveau de l’IC1
Y1ech=((Xbar-quantile*sqrt(Xbar)/sqrt(n))<lambda)...
&((Xbar+quantile*sqrt(Xbar)/sqrt(n))>lambda);

niveauIC1=[niveauIC1,mean(1*Y1ech)]

//niveau de l’IC1bis
Y1bisech=((Xbar-quantile*sqrt(S2)/sqrt(n))<lambda)...
&((Xbar+quantile*sqrt(S2)/sqrt(n))>lambda);

niveauIC1bis=[niveauIC1bis,mean(1*Y1bisech)]

//niveau de l’IC2
Y2ech=(((sqrt(Xbar)-quantile/(2*sqrt(n))).^2)<lambda)...
&(((sqrt(Xbar)+quantile/(2*sqrt(n))).^2)>lambda);

niveauIC2=[niveauIC2,mean(1*Y2ech)]
end
disp(niveauIC1)
disp(niveauIC1bis)
disp(niveauIC2)

